
Abstract

Given a string of parentheses, the task is to find the longest consecutive segment that is balanced,
in linear time. We find this problem interesting because it involves a combination of techniques:
the usual approach for solving segment problems and a theorem for constructing the inverse of a
function—through which we derive an instance of shift-reduce parsing.

1 Introduction

Given a string of parentheses, the task is to find a longest consecutive segment that is
balanced. For example, for input "))(()())())()(" the output should be "(()())()".
We also consider a reduced version of the problem in which we return only the length
of the segment. While there is no direct application of this problem,1 the authors find it
interesting because it involves two techniques. Firstly, derivation for such optimal segment
problems (those whose goal is to compute a segment of a list that is optimal up to certain
criteria) usually follows a certain pattern (e.g. Bird 1987; Zantema 1992; Gibbons 1997).
We would like to see how well that works for this case. Secondly, at one point, we will
need to construct the right inverse of a function. It will turn out that we will discover an
instance of shift-reduce parsing.

Specification Balanced parentheses can be captured by a number of grammars, for exam-
ple, S→ ε | (S) | SS, or S→T∗ and T→ (S). After trying some of them, the authors
decided on

S→ ε | (S) S,

1 However, the length-only version was possibly used as an interview problem collected in, for example,
https://leetcode.com/problems/longest-valid-parentheses/.

because it is unambiguous and the most concise. Other grammars have worked too, albeit
leading to lengthier algorithms. The parse tree of the chosen grammar can be represented
in Haskell as below, with a function pr specifying how a tree is printed:

data Tree=Nul |Bin Tree Tree,

pr :: Tree→ String
pr Nul = ""
pr (Bin t u)= "("++ pr t++ ")"++ pr u.

For example, letting t1 =Bin Nul Nul and t2 =Bin Nul (Bin Nul Nul), we have pr t1 =
"()", pr t2 = "()()" and pr (Bin t2 t1)= "(()())()" (parentheses are coloured to aid
the reader).

Function pr is injective but not surjective: it does not yield unbalanced strings. Therefore
its right inverse, that is, the function pr−1 such that pr (pr−1 xs)= xs, is partial; its domain
is the set of balanced parenthesis strings. We implement it by a function that is made total
by using the Maybe monad. This function parse :: String→Maybe Tree builds a parse
tree—parse xs should return Just t such that pr t= xs if xs is balanced and return Nothing
otherwise. While this defines parse already, a direct definition of parse will be presented
in Section 4.

The problem can then be specified as below, where lbs stands for “longest balanced
segment (of parentheses)”:

lbs :: String→Tree
lbs=maxBy size · filtJust ·map parse · segments,

segments= concat ·map inits · tails,
filtJust ts= [ t | Just t← ts],
size t= length (pr t).

The function segments :: [a]→ [[a]] returns all segments of a list, with inits, tails :: [a]→
[[a]], respectively, computing all prefixes and suffixes of their input lists. The result of
map parse is passed to filtJust :: [Maybe a]→ [a], which collects only those elements
wrapped by Just. For example, filtJust [Just 1, Nothing, Just 2]= [1, 2].2 For this prob-
lem filtJust always returns a non-empty list, because the empty string, which is a member
of segments xs for any xs, can always be parsed to Just Nul. Given f :: a→ b where b is a
type that is ordered, maxBy f :: [a]→ a picks a maximum element from the input. Finally,
size t computes the length of pr t.

The length-only problem can be specified by lbsl= size · lbs.

2 The prefix–suffix decomposition

It is known that many optimal segment problems can be solved by following a fixed pattern
(Bird, 1987; Zantema, 1992; Gibbons, 1997), which we refer to as prefix–suffix decompo-
sition. In the first step, finding an optimal segment is factored into finding, for each suffix,
an optimal prefix. For our problem, the calculation goes

2 filtJust is called catMaybes in the standard Haskell libraries. The authors think the name filtJust is more
informative.

https://doi.org/10.1017/S0956796821000253 Published online by Cambridge University Press

https://doi.org/10.1017/S0956796821000253


Longest segment of balanced parentheses 3

maxBy size · filtJust ·map parse · segments
= { definition of segments }

maxBy size · filtJust ·map parse · concat ·map inits · tails
= { since map f · concat= concat ·map (map f ), map fusion }

maxBy size · filtJust · concat ·map (map parse · inits) · tails
= { since filtJust · concat= concat ·map filtJust }

maxBy size · concat ·map (filtJust ·map parse · inits) · tails
= { since maxBy f · concat=maxBy f ·map (maxBy f ) }

maxBy size ·map (maxBy size · filtJust ·map parse · inits) · tails.

For each suffix returned by tails, the program above computes its longest prefix of balanced
parentheses by maxBy size · filtJust ·map parse · inits. We abbreviate the latter to lbp (for
“longest balanced prefix”).

Generating every suffix and computing lbp for each of them is rather costly. The next
step is to try to apply the following scan lemma, which says that if a function f can be
expressed as right fold, there is a more efficient algorithm to compute map f · tails:

Lemma 2.1. map (foldr (⊕) e) · tails= scanr (⊕) e, where

scanr :: (a→ b→ b)→ b→ [a]→ [b]
scanr (⊕) e [ ] = [e]
scanr (⊕) e (x : xs)= let (y : ys)= scanr (⊕) e xs in (x⊕ y) : y : ys.

If lbp can be written in the form foldr (⊕) e, we do not need to compute lbp of each
suffix from scratch; each optimal prefix can be computed, in scanr, from the previous
optimal prefix by (⊕). If (⊕) is a constant-time operation, we get a linear-time algorithm.

The next challenge is therefore to express lbp as a right fold. Since inits can be expressed
as a right fold—inits= foldr (λx xss→ [ ] : map (x:) xss) [[ ]], a reasonable attempt is to
fuse maxBy size · filtJust ·map parse with inits, to form a single foldr. Recall the foldr-
fusion theorem:

Theorem 2.2 (foldr-fusion). h · foldr f e= foldr g (h e) if h (f x y)= g x (h y).

The antecedent h (f x y)= g x (h y) will be referred to as the fusion condition. To fuse
map parse and inits using Theorem 2.2, we calculate from the LHS of the fusion condition
(with h=map parse and f = (λx xss→ [ ] : map (x:) xss)):

map parse ([ ] : map (x:) xss)
= { since parse [ ]= Just Nul }

Just Nul : map (parse · (x:)) xss
= { wish, for some g′ }

Just Nul : g′ x (map parse xss)
= { let g x ts= Just Nul : g x ts }

g x (map parse xss).

We can construct g if (and only if) there is a function g′ such that g′ x (map parse xss)=
map (parse · (x:)) xss. Is that possible?

It is not hard to see that the answer is no. Consider xss= [")", ")()"] and x= ’(’.
Since both strings in xss are not balanced, map parse xss gives us [Nothing, Nothing].

However, map (x:) xss= ["()", "()()"], a list of balanced strings. Therefore, g′ has to
produce something from nothing—an impossible task. We have to generalise our problem
such that g′ receives inputs that are more informative.

3 Partially balanced strings

A string of parentheses is said to be left-partially balanced if it may possibly be bal-
anced by adding zero or more parentheses to the left. For example, xs="(())()))()" is
left-partially balanced because "(("++ xs is balanced—again we use colouring to help the
reader parsing the string. Note that "(()(" ++ xs is also balanced. For a counter example,
the string ys= "()(()" is not left-partially balanced—due to the unmatched ’(’ in the
middle of ys, there is no zs such that zs++ ys can be balanced.

While parsing a fully balanced string cannot be expressed as a right fold, it is possible
to parse left-partially balanced strings using a right fold. In this section, we consider what
data structure such a string should be parsed to. We discuss how to how to parse it in the
next section.

A left-partially balanced string can always be uniquely factored into a sequence of fully
balanced substrings, separated by one or more right parentheses. For example, xs can be
factored into two balanced substrings, "(())()" and "()", separated by "))". One of the
possible ways to represent such a string is by a list of trees—a Forest, where the trees are
supposed to be separated by a ’)’. That is, such a forest can be printed by:

type Forest= [Tree], {− non-empty −}
prF :: Forest→ String
prF [ t ] = pr t
prF (t : ts)= pr t++ ")"++ prF ts.

For example, xs= "(())()))()" can be represented by a forest containing three trees:

ts= [Bin (Bin Nul Nul) (Bin Nul Nul), Nul, Bin Nul Nul],

where Bin (Bin Nul Nul) (Bin Nul Nul) prints to "(())()", Bin Nul Nul prints to "()", and
there is a Nul between them due to the consecutive right parentheses "))" in xs (Nul itself
prints to ""). One can verify that prF ts= xs indeed. Note that we let the type Forest be
non-empty lists of trees.3 The empty string can be represented by [Nul], since prF [Nul]=
pr Nul= "".

The aim now is to construct the right inverse of prF, such that a left-partially balanced
string can be parsed using a right fold.

4 Parsing partially balanced strings

Given a function f :: b→ t, the converse-of-a-function theorem (Bird & de Moor, 1997;
de Moor & Gibbons, 2000) constructs the relational converse—a generalised notion of
inverse—of f . The converse is given as a relational fold whose input type is t, which can

3 We can let the non-emptiness be more explicit by letting Forest= (Tree, [Tree]). Presentation-wise, both
representations have their pros and cons, and we eventually decided on using a list.

be any inductively defined datatype with a polynomial base functor. We specialise the
general theorem to our needs: we use it to construct only functions, not relations, and only
for the case where t is a list type.

Theorem 4.1. Given f :: b→ [a], if we have base :: b and step :: a→ b→ b satisfying:

f base= [ ] ∧
f (step x t)= x : f t,

then f −1 = foldr step base is a partial right inverse of f . That is, we have f (f −1 xs)= xs for
all xs in the range of f .

While the general version of the theorem is not trivial to prove, the version above,
specialised to functions and lists, can be verified by an easy induction on the input list.

Recall that we wish to construct the right inverse of prF using Theorem 4.1. It will
be easier if we first construct a new definition of prF, one that is inductive, does not use
(++) and does not rely on pr. For a base case, prF [Nul]= "". It is also immediate that
prF (Nul : ts)= ’)’ : prF ts. When the list contains more than one tree and the first tree is
not Nul, we calculate

prF (Bin t u : ts)
= { definitions of pr and prF }
"("++ pr t++ ")"++ pr u++ ")"++ prF ts
= { definition of prF }
’(’ : prF (t : u : ts).

We have thus derived the following new definition of prF:

prF [Nul] = ""
prF (Nul : ts) = ’)’ : prF ts
prF (Bin t u : ts)= ’(’ : prF (t : u : ts).

We are now ready to invert prF by Theorem 4.1, which amounts to finding base and
step such that prF base= "" and prF (step x ts)= x : prF ts for x= ’(’ or ’)’. With the
inductive definition of prF in mind, we pick base= [Nul], and the following step meets
the requirement:

step ’)’ ts =Nul : ts
step ’(’ (t : u : ts)=Bin t u : ts.

We have thus constructed prF−1 = foldr step [Nul]. If we expand the definitions, we have

prF−1 :: String→ Forest
prF−1 "" = [Nul]
prF−1 (’)’ : xs)=Nul : prF−1 xs
prF−1 (’(’ : xs)= case prF−1 xs of (t : u : ts)→Bin t u : ts,

which is pleasingly symmetrical to the inductive definition of prF.
For an operational explanation, a right parenthesis ’)’ indicates starting a new tree,

thus we start freshly with a Nul; a left parenthesis ’(’ ought to be the leftmost symbol
of some "(t)u", thus we wrap the two most recent siblings into one tree. When there are

no such two siblings (i.e. prF−1 xs in the case expression evaluates to a singleton list), the
input ’(’ : xs is not a left-partially balanced string—’(’ appears too early, and the result
is undefined.

Readers may have noticed the similarity to shift-reduce parsing, in which, after reading
a symbol we either “shift” the symbol by pushing it onto a stack, or “reduce” the symbol
against a top segment of the stack. Here, the forest is the stack. The input is processed right-
to-left, as opposed to left-to-right, which is more common when talking about parsing. We
shall discuss this issue further in Section 7.

We could proceed to work with prF−1 for the rest of this pearl but, for clarity, we
prefer to make the partiality explicit. Let parseF be the monadified version of prF−1,
given by:

parseF :: String→Maybe Forest
parseF "" = Just [Nul]
parseF (x : xs)= parseF xs >>= stepM x,

where stepM ’)’ ts = Just (Nul : ts)
stepM ’(’ [ t ] =Nothing
stepM ’(’ (t : u : ts)= Just (Bin t u : ts),

where stepM is monadified step—for the case [ t ] missing in step we return Nothing.
To relate parseF to parse, notice that prF [ t ]= pr t. We therefore have

parse :: String→Maybe Tree
parse= unwrapM <=< parseF,

unwrapM [ t ]= Just t
unwrapM =Nothing.

where (<=<) :: (b→M c)→ (a→M b)→ (a→M c) is (reversed) Kleisli composition.
That is, parse calls parseF and declares success only when the input can be parsed into a
single tree.

5 Longest balanced prefix in a fold

Recall our objective at the close of Section 2: to compute lbp=maxBy size · filtJust ·
map parse · inits in a right fold in order to obtain a faster algorithm using the scan lemma.
Now that we have parse= unwrapM <=< parseF where parseF is a right fold, and we per-
form some initial calculation whose purpose is to factor the post-processing unwrapM out
of the main computation:

maxBy size · filtJust ·map parse · inits
= { since parse= unwrapM <=< parseF }

maxBy size · filtJust ·map (unwrapM <=< parseF) · inits
= { since (f <=< g) x= f =<< g x, map fusion (backwards) }

maxBy size · filtJust ·map (unwrapM=<<) ·map parseF · inits
= { since filtJust ·map (unwrapM=<<)=map unwrap · filtJust, see below }

maxBy size ·map unwrap · filtJust ·map parseF · inits
= { since maxBy f ·map g= g ·maxBy (f · g) }

unwrap ·maxBy (size · unwrap) · filtJust ·map parseF · inits.

In the penultimate step (unwrapM=<<) is moved leftwards past filtJust and becomes
unwrap :: Forest→Tree, defined by:

unwrap [ t ]= t
unwrap =Nul.

Recall that inits= foldr (λx xss→ [ ] : map (x:) xss) [[ ]]. The aim now is to fuse
map parseF, filtJust and maxBy (size · unwrap) with inits.

By Theorem 2.2, to fuse map parseF with inits, we need to construct g that meets the
fusion condition:

map parseF ([ ] : map (x:) xss)= g x (map parseF xss).

Now that we know that parseF is a fold, the calculation goes

map parseF ([ ] : map (x:) xss)
= { definitions of map and parseF }

Just [Nul] : map (parseF · (x:)) xss
= { the foldr definition of parseF }

Just [Nul] : map (λts→ parseF ts >>= stepM x) xss
= { map-fusion (backwards) }

Just [Nul] : map (>>= stepM x) (map parseF xss).

Therefore, we have

map parseF · inits :: String→ [Maybe Forest]
map parseF · inits=

foldr (λx tss→ Just [Nul] : map (>>= stepM x) tss) [Just [Nul]].

Next, we fuse filtJust with map parseF · inits by Theorem 2.2. After some calculations,
we get

filtJust ·map parseF · inits :: String→ [Forest]
filtJust ·map parseF · inits= foldr (λx tss→ [Nul] : extend x tss) [[Nul]],

where extend ’)’ tts=map (Nul:) tts
extend ’(’ tts= [(Bin t u : ts) | (t : u : ts)← tts].

After the fusion, we need not keep the Nothing entries in the fold; the computation returns
a collection of forests. If the next character is ’)’, we append Nul to every forest. If the
next entry is ’(’, we choose those forests having at least two trees and combine them—the
list comprehension keeps only the forests that match the pattern (t : u : ts) and throws away
those do not. Note that [Nul], to which the empty string is parsed, is always added to the
collection of forests.

To think about how to deal with unwrap ·maxBy (size · unwrap), we consider an exam-
ple. Figure 1 shows the results of map parseF and filtJust for prefixes of "())()(",
where Just, Nul and Bin are, respectively, abbreviated to J, N and B. The function
maxBy (size · unwrap) chooses between [N] and [B N N], the two parses resulting in single
trees, and returns [B N N]. However, notice that B N N is also the head of [B N N, B N N],
the last forest returned by filtJust. In general, the largest singleton parse tree will also
present in the head of the last forest returned by filtJust ·map parseF · inits. One can

Fig. 1: Results of parseF and filtJust for prefixes of "())()(".

intuitively see why: if we print them both, the former is a prefix of the latter. Therefore,
unwrap ·maxBy (size · unwrap) can be replaced by head · last.

To fuse last with filtJust ·map parseF · inits by Theorem 2.2, we need to construct a
function step that satisfies the fusion condition:

last ([Nul] : extend x tss)= step x (last tss),

where tss is a non-empty list of forests. The case when x= ’)’ is easy—choosing
step ’)’ ts=Nul : ts will do the job. For the case when x= ’(’ we need to analyse the
result of last tss and use the property that forests in tss are ordered in ascending lengths.
(a) If last tss= [ t ], a forest having only one tree, there are no forest in tss that contains

two or more trees. Therefore, extend ’(’ tss returns an empty list, and last ([Nul] :
extend ’(’ tss)= [Nul].

(b) Otherwise, extend ’(’ tss would not be empty, and last ([Nul] : extend x tss)=
last (extend x tss). We may then combine the first two trees, as extend would do.

In summary, we have

last · filtJust ·map parseF · inits :: String→ Forest
last · filtJust ·map parseF · inits= foldr step [Nul],

where step ’)’ ts =Nul : ts
step ’(’ [ t ] = [Nul]
step ’(’ (t : u : ts)=Bin t u : ts,

which is now a total function on strings of parentheses.
The function derived above turns out to be prF−1 with one additional case (step ’(’ [ t ]=

[Nul]). What we have done in this section can be seen as justifying this extra case (which
is a result of case (1) in the fusion of last), which is not as trivial as one might think.

6 Wrapping up

We can finally resume the main derivation in Section 2:

maxBy size ·map (maxBy size · filtJust ·map parse · inits) · tails
= { Section 5: lbp= head · foldr step [Nul] }

maxBy size ·map (head · foldr step [Nul]) · tails
= { map-fusion reversed, Lemma 2.1 }

maxBy size ·map head · scanr step [Nul].

https://doi.org/10.1017/S0956796821000253 Published online by Cambridge University Press

https://doi.org/10.1017/S0956796821000253


Longest segment of balanced parentheses 9

Fig. 2: Measured running time for some input sizes.

We have therefore derived:

lbs :: String→Tree
lbs=maxBy size ·map head · scanr step [Nul],

where step is as defined in the end of Section 5. To avoid recomputing the sizes in
maxBy size, we can annotate each tree by its size: letting Forest= [(Tree, Int)], resulting
in an algorithm that runs in linear time:

lbs :: String→Tree
lbs= fst ·maxBy snd ·map head · scanr step [(Nul, 0)],

where step ’)’ ts = (Nul, 0) : ts
step ’(’ [ t ]= [(Nul, 0)]
step ’(’ ((t, m) : (u, n) : ts)= (Bin t u, 2+m+ n) : ts.

Finally, the size-only version can be obtained by fusing size into lbs. It turns out that we
do not need to keep the actual trees, but only their sizes—Forest= [Int]:

lbsl :: String→ Int
lbsl=maximum ·map head · scanr step [0],

where step ’)’ ts = 0 : ts
step ’(’ [ t ]= [0]
step ’(’ (m : n : ts)= (2+m+ n) : ts.

We ran some simple experiments to measure the efficiency of the algorithm. The test
machine was a laptop computer with a Apple M1 chip (8 core, 3.2GHz) and 16GB RAM.
We ran lbs on randomly generated inputs containing 1, 2, 4, 6, 8 and 10 million parentheses
and measured the user times. The results, shown in Figure 2, confirmed the linear-time
behaviour.

7 Conclusions and discussions

So we have derived a linear-time algorithm for solving the problem. We find it an inter-
esting journey because it relates two techniques: prefix–suffix decomposition for solving
segment problems and the converse-of-a-function theorem for program inversion.

In Section 3, we generalised from trees to forests. Generalisations are common when
applying the converse-of-a-function theorem. It was observed that the trees in a forest are
those along the left spine of the final tree; therefore, such a generalisation is referred to as
switching to a “spine representation” (Mu & Bird, 2003).

What we derived in Sections 4 and 5 is a compacted form of shift-reduce parsing, where
the input is processed right to left. The forest serves as the stack, but we do not need
to push the parser state to the stack, as is done in shift-reduce parsing. If we were to
process the input in the more conventional left-to-right order, the corresponding grammar

would be S→ ε | S (S). It is an SLR(1) grammar whose parse table contains five states.
Our program is much simpler. A possible reason is that consecutive shifting and reducing
are condensed into one step. It is likely that parsing SLR(1) languages can be done in a
fold. The relationship between LR parsing and the converse-of-a-function theorem awaits
further investigation.

There are certainly other ways to solve the problem. For example, one may interpret a
’(’ as a−1, and a ’)’ as a+1. A left-partially balanced string would be a list whose right-
to-left running sum is never negative. One may then apply the method in Zantema (1992) to
find the longest such prefix for each suffix. The result will be an algorithm that maintains
the sum in a loop—an approach that might be more commonly adopted by imperative
programmers. The problem can also be seen as an instance of maximum-marking prob-
lems—choosing elements in a data structure that meet a given criteria while maximising a
cost function—to which methods of Sasano et al. (2001) can be applied.

